

NAG Fortran Library Routine Document

E04UQF/E04UQA

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

To supply optional parameters to E04USF/E04USA from an external file. More precisely, E04UQF must be used to supply optional parameters to E04USF and E04UQA must be used to supply optional parameters to E04USA.

E04UQA is a version of E04UQF that has additional parameters in order to make it safe for use in multithreaded applications (see Section 5 below). The initialisation routine E04WBF **must** have been called prior to calling E04UQA.

2 Specifications

2.1 Specification for E04UQF

```
SUBROUTINE E04UQF(IOPTNS, INFORM)
  INTEGER          IOPTNS, INFORM
```

2.2 Specification for E04UQA

```
SUBROUTINE E04UQA(IOPTNS, LWSAV, IWSAV, RWSAV, INFORM)
  INTEGER          IOPTNS, IWSAV(610), INFORM
  real            RWSAV(475)
  LOGICAL          LWSAV(120)
```

3 Description

E04UQF/E04UQA may be used to supply values for optional parameters to the corresponding routines E04USF/E04USA. E04UQF/E04UQA reads an external file and each line of the file defines a single optional parameter. It is only necessary to supply values for those parameters whose values are to be different from their default values.

Each optional parameter is defined by a single character string, of up to 72 characters, consisting of one or more items. The items associated with a given option must be separated by spaces, or equals signs [=]. Alphabetic characters may be upper or lower case. The string

```
Print level = 1
```

is an example of a string used to set an optional parameter. For each option the string contains one or more of the following items:

- (a) A mandatory keyword.
- (b) A phrase that qualifies the keyword.
- (c) A number that specifies an INTEGER or **real** value. Such numbers may be up to 16 contiguous characters in Fortran's I, F, E or D formats, terminated by a space if this is not the last item on the line.

Blank strings and comments are ignored. A comment begins with an asterisk (*) and all subsequent characters in the string are regarded as part of the comment.

The file containing the options must start with **begin** and must finish with **end**. An example of a valid options file is:

```
Begin * Example options file
  Print level = 5
End
```

For E04UQF each line of the file is normally printed as it is read, on the current advisory message unit (see X04ABF), but printing may be suppressed using the keyword **nolist**. To suppress printing of **begin**, **nolist** must be the first option supplied as in the file:

```
Begin
  Nolist
  Print level = 5
End
```

Printing will automatically be turned on again after a call to E04USF/E04USA or E04UQF and may be turned on again at any time using the keyword **list**.

For E04UQA printing is turned off by default, but may be turned on at any time using the keyword **list**.

Optional parameter settings are preserved following a call to E04USF/E04USA and so the keyword **defaults** is provided to allow you to reset all the optional parameters to their default values prior to a subsequent call to E04USF/E04USA.

A complete list of optional parameters, their abbreviations, synonyms and default values is given in Section 11 of the document for E04UNF or Section 11 of the document for E04USF/E04USA.

4 References

None.

5 Parameters

1: IOPTNS – INTEGER *Input*

On entry: the unit number of the options file to be read.

Constraint: $0 \leq \text{IOPTNS} \leq 99$.

2: INFORM – INTEGER *Output*

Note: for E04UQA, INFORM does not occur in this position in the parameter list. See the additional parameters described below.

On exit: contains zero if the options file has been successfully read and a value > 0 otherwise (see Section 6).

Note: the following are additional parameters for specific use with E04UQA. Users of E04UQF therefore need not read the remainder of this section.

2: LWSAV(120) – LOGICAL array *Workspace*

3: IWSAV(610) – INTEGER array *Workspace*

4: RWSAV(475) – *real* array *Workspace*

The arrays LWSAV, IWSAV and RWSAV **must not** be altered between calls to any of the routines E04WBF, E04USA, E04UQA or E04URA.

5: INFORM – INTEGER *Output*

On exit: contains zero if the options file has been successfully read and a value > 0 otherwise (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFORM = 1

IOPTNS is not in the range [0,99].

INFORM = 2

begin was found, but end-of-file was found before **end** was found.

INFORM = 3

end-of-file was found before **begin** was found.

INFORM = 4

Not used.

INFORM = 5

One or more lines of the options file is invalid. Check that all keywords are neither ambiguous nor misspelt.

7 Accuracy

Not applicable.

8 Further Comments

E04URF/E04URA may also be used to supply optional parameters to the corresponding routines E04USF/E04USA.

9 Example

This example solves the same problem as the example for E04USF/E04USA, but in addition illustrates the use of E04UQF/E04UQA and E04URF/E04URA to set optional parameters for E04USF/E04USA.

In this example the options file read by E04UQF/E04UQA is appended to the data file for the program (see E04UQF/E04UQA). It would usually be more convenient in practice to keep the data file and the options file separate.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

Note: the following program illustrates the use of E04UQF. An equivalent program illustrating the use of E04UQA is available with the supplied Library and is also available from the NAG web site.

```
*      E04UQF Example Program Text
*      Mark 20 Revised. NAG Copyright 2001.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER        (NIN=5,NOUT=6)
INTEGER          MMAX, NMAX, NCLMAX, NCNMAX
PARAMETER        (MMAX=50,NMAX=10,NCLMAX=10,NCNMAX=10)
INTEGER          LDA, LDCJ, LDFJ, LDR
PARAMETER        (LDA=NCLMAX,LDCJ=NCNMAX,LDFJ=MMAX,LDR=NMAX)
INTEGER          LIWORK, LWORK
PARAMETER        (LIWORK=100,LWORK=1000)
*      .. Local Scalars ..
real            OBJF
```

```

      INTEGER          I, IFAIL, INFORM, ITER, J, M, N, NCLIN, NCNLN
*    .. Local Arrays ..
      real             A(LDA,NMAX), BL(NMAX+NCLMAX+NCNMAX),
+                     BU(NMAX+NCLMAX+NCNMAX), C(NCNMAX),
+                     CJAC(LDCJ,NMAX), CLAMDA(NMAX+NCLMAX+NCNMAX),
+                     F(MMAX), FJAC(LDFJ,NMAX), R(LDR,NMAX), USER(1),
+                     WORK(LWORK), X(NMAX), Y(MMAX)
      INTEGER          ISTATE(NMAX+NCLMAX+NCNMAX), IUSER(1),
+                     IWORK(LIWORK)
*    .. External Subroutines ..
      EXTERNAL         CONFUN, E04UQF, E04URF, E04USF, OBJFUN, X04ABF
*    .. Executable Statements ..
      WRITE (NOUT,*) 'E04UQF Example Program Results'
*    Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      READ (NIN,*) NCLIN, NCNLN
      IF (M.LE.MMAX .AND. N.LE.NMAX .AND. NCLIN.LE.NCLMAX .AND.
+       NCNLN.LE.NCNMAX) THEN
*
*       Read A, Y, BL, BU and X from data file
*
      IF (NCLIN.GT.0) READ (NIN,*) ((A(I,J),J=1,N),I=1,NCLIN)
      READ (NIN,*) (Y(I),I=1,M)
      READ (NIN,*) (BL(I),I=1,N+NCLIN+NCNLN)
      READ (NIN,*) (BU(I),I=1,N+NCLIN+NCNLN)
      READ (NIN,*) (X(I),I=1,N)
*
*       Set three options using E04URF
*
      CALL E04URF(' Infinite Bound Size = 1.0D+25 ')
*
      CALL E04URF(' Print Level = 1 ')
*
      CALL E04URF(' Verify Level = -1 ')
*
*       Set the unit number for advisory messages to NOUT
*
      CALL X04ABF(1,NOUT)
*
*       Read the options file for the remaining options
*
      CALL E04UQF(NIN,INFORM)
*
      IF (INFORM.NE.0) THEN
+       WRITE (NOUT,99999) 'E04UQF terminated with INFORM = ',
+       INFORM
+       STOP
      END IF
*
*       Solve the problem
*
      IFAIL = -1
*
      CALL E04USF(M,N,NCLIN,NCNLN,LDA,LDCJ,LDFJ,LDR,A,BL,BU,Y,CONFUN,
+       OBJFUN,ITER,ISTATE,C,CJAC,F,FJAC,CLAMDA,OBJF,R,X,
+       IWORK,LIWORK,WORK,LWORK,IUSER,USER,IFAIL)
*
      END IF
      STOP
*
99999 FORMAT (1X,A,I3)
      END
      SUBROUTINE OBJFUN(MODE,M,N,LDFJ,NEEDFI,X,F,FJAC,NSTATE,IUSER,USER)
*       Routine to evaluate the subfunctions and their 1st derivatives.
*       .. Parameters ..
      real             PT49, ONE, EIGHT
      PARAMETER        (PT49=0.49e0,ONE=1.0e0,EIGHT=8.0e0)
*       .. Scalar Arguments ..
      INTEGER          LDFJ, M, MODE, N, NEEDFI, NSTATE
*       .. Array Arguments ..

```

```

      real          F(*), FJAC(LDFJ,*), USER(*), X(N)
      INTEGER      IUSER(*)
*
* .. Local Scalars ..
      real          AI, TEMP, X1, X2
      INTEGER      I
      LOGICAL      MODE02, MODE12
*
* .. Local Arrays ..
      real          A(44)
*
* .. Intrinsic Functions ..
      INTRINSIC      EXP
*
* .. Data statements ..
      DATA          A/8.0e0, 8.0e0, 10.0e0, 10.0e0, 10.0e0, 10.0e0,
+                   12.0e0, 12.0e0, 12.0e0, 12.0e0, 14.0e0, 14.0e0,
+                   14.0e0, 16.0e0, 16.0e0, 16.0e0, 18.0e0, 18.0e0,
+                   20.0e0, 20.0e0, 20.0e0, 22.0e0, 22.0e0, 22.0e0,
+                   24.0e0, 24.0e0, 24.0e0, 26.0e0, 26.0e0, 26.0e0,
+                   28.0e0, 28.0e0, 30.0e0, 30.0e0, 30.0e0, 32.0e0,
+                   32.0e0, 34.0e0, 36.0e0, 36.0e0, 38.0e0, 38.0e0,
+                   40.0e0, 42.0e0/
*
* .. Executable Statements ..
      X1 = X(1)
      X2 = X(2)
      MODE02 = MODE .EQ. 0 .OR. MODE .EQ. 2
      MODE12 = MODE .EQ. 1 .OR. MODE .EQ. 2
      DO 20 I = 1, M
        IF (NEEDFI.EQ.I) THEN
          F(I) = X1 + (PT49-X1)*EXP(-X2*(A(I)-EIGHT))
          RETURN
        ELSE
          AI = A(I)
          TEMP = EXP(-X2*(AI-EIGHT))
          IF (MODE02) F(I) = X1 + (PT49-X1)*TEMP
          IF (MODE12) THEN
            FJAC(I,1) = ONE - TEMP
            FJAC(I,2) = -(PT49-X1)*(AI-EIGHT)*TEMP
          END IF
        END IF
      END IF
20 CONTINUE
*
      RETURN
      END
*
      SUBROUTINE CONFUN(MODE,NCNLN,N,LDCJ,NEEDC,X,C,CJAC,NSTATE,IUSER,
+                      USER)
*
* Routine to evaluate the nonlinear constraint and its 1st
* derivatives.
* .. Parameters ..
      real          ZERO, PT09, PT49
      PARAMETER      (ZERO=0.0e0,PT09=0.09e0,PT49=0.49e0)
*
* .. Scalar Arguments ..
      INTEGER      LDCJ, MODE, N, NCNLN, NSTATE
*
* .. Array Arguments ..
      real          C(*), CJAC(LDCJ,*), USER(*), X(N)
      INTEGER      IUSER(*), NEEDC(*)
*
* .. Local Scalars ..
      INTEGER      I, J
*
* .. Executable Statements ..
      IF (NSTATE.EQ.1) THEN
        First call to CONFUN. Set all Jacobian elements to zero.
        Note that this will only work when 'Derivative Level = 3'
        (the default; see Section 11.2).
        DO 40 J = 1, N
          DO 20 I = 1, NCNLN
            CJAC(I,J) = ZERO
          20 CONTINUE
        40 CONTINUE
      END IF
*
      IF (NEEDC(1).GT.0) THEN
        IF (MODE.EQ.0 .OR. MODE.EQ.2) C(1) = -PT09 - X(1)*X(2) +
+          PT49*X(2)

```

```

      IF (MODE.EQ.1 .OR. MODE.EQ.2) THEN
        CJAC(1,1) = -X(2)
        CJAC(1,2) = -X(1) + PT49
      END IF
    END IF
  *
  RETURN
END

```

9.2 Program Data

E04UQF Example Program Data

```

44  2                                :Values of M and N
1   1                                :Values of NCLIN and NCNLN
1.0 1.0                             :End of matrix A
0.49 0.49 0.48 0.47 0.48 0.47 0.46 0.46 0.45 0.43 0.45
0.43 0.43 0.44 0.43 0.43 0.46 0.45 0.42 0.42 0.43 0.41
0.41 0.40 0.42 0.40 0.40 0.41 0.40 0.41 0.41 0.40 0.40
0.40 0.38 0.41 0.40 0.40 0.41 0.38 0.40 0.40 0.39 0.39 :End of Y
0.4 -4.0 1.0 0.0 :End of BL
1.0E+25 1.0E+25 1.0E+25 1.0E+25 :End of BU
0.4 0.0 :End of X
Begin Example options file for E04UQF
  Major Iteration Limit = 15 * (Default = 50)
  Minor Iteration Limit = 10 * (Default = 50)
End

```

9.3 Program Results

E04UQF Example Program Results

Calls to E04URF

```

Infinite Bound Size = 1.0E+25
Print Level = 1
Verify Level = -1

```

OPTIONS file

```

Begin Example options file for E04UQF
  Major Iteration Limit = 15 * (Default = 50)
  Minor Iteration Limit = 10 * (Default = 50)
End

```

*** E04USF

*** Start of NAG Library implementation details ***

```

Implementation title: Generalised Base Version
Precision: FORTRAN double precision
Product Code: FLBAS20D
Mark: 20A

```

*** End of NAG Library implementation details ***

Parameters

Linear constraints.....	1	Variables.....	2
Nonlinear constraints..	1	Subfunctions.....	44
Infinite bound size....	1.00E+25	COLD start.....	
Infinite step size.....	1.00E+25	EPS (machine precision)	1.11E-16
Step limit.....	2.00E+00	Hessian.....	NO
Linear feasibility.....	1.05E-08	Crash tolerance.....	1.00E-02
Nonlinear feasibility..	1.05E-08	Optimality tolerance...	3.26E-12

```

Line search tolerance.. 9.00E-01      Function precision..... 4.38E-15
Derivative level.....      3      Monitoring file.....      -1
Verify level.....      -1
Major iterations limit.      15      Major print level.....      1
Minor iterations limit.      10      Minor print level.....      0
J'J initial Hessian....      Reset frequency.....      2

Workspace provided is      IWORK(      100), WORK(      1000).
To solve problem we need IWORK(      9), WORK(      306).

Exit from NP problem after      6 major iterations,
                                8 minor iterations.

```

Varbl	State	Value	Lower Bound	Upper Bound	Lagr Mult	Slack
V	1	FR	0.419953	0.400000	None	1.9953E-02
V	2	FR	1.28485	-4.00000	None	5.285

L	Con State	Value	Lower Bound	Upper Bound	Lagr Mult	Slack
L	1	FR	1.70480	1.00000	None	0.7048

N	Con State	Value	Lower Bound	Upper Bound	Lagr Mult	Slack
N	1	LL	-9.768135E-13	.	None	3.3358E-02 -9.7681E-13

Exit E04USF - Optimal solution found.

Final objective value = 0.1422983E-01
